A Rolling Square Bridge

Antonín Slavík, Charles University, Prague, Czech Republic; slavik@karlin.mff.cuni.cz

Stan Wagon, Macalester College, St. Paul, Minnesota, USA; wagon@macalester.edu

Introduction

Back in January 1991 Wolfram Research organized the first *Mathematica* conference, in Redwood City, California. The conference banquet was held at San Francisco's *Exploratorium* and there Leon Hall (Univ. of Missouri, Rolla) and S.W. saw a model of a square wheel rolling smoothly on a certain bumpy road. This was completely new to them, so later that evening they borrowed a computer and exercised their mathematical and *Mathematica*'l skills to work out the geometry of the construction and make an animation of a square rolling along a sequence of catenaries. That led to some interesting large models: first a full-size tricycle that can be ridden smoothly on a 25-foot road, and more recently a 26,000-pound bridge that can be rolled, by hand, so that boats can pass underneath it. The ideas underlying this bridge design are so radical, nontraditional, and elegant that the bridge won the Bridges Design Award for 2022 [11].

An Elegant Differential Equation

Suppose a wheel is given in polar coordinates as $r = r(\theta)$, with the origin viewed as the center of mass, and the corresponding road (i.e, the road on which the wheel will roll—we assume friction prevents any sliding—with the center of mass moving neither up nor down) is $y = f(x)$; see Figure 1 and note that y is negative.Let $\theta(x)$ be the polar angle defining the point that will touch the road at x.

Out[ ]=



Figure 1. The relationship between a wheel and the corresponding road. Here the road is $\cos(x) - \sqrt{17}$ and the

wheel's polar form is $r = \sqrt{17} - \frac{9+\sqrt{17}-8\tan^2(2\theta)}{9+\sqrt{17}+8\tan^2(2\theta)}$.

We have the *radius condition*: $f(x) = -r(\theta(x))$. And matching the arc length of the road to the relevant part of the perimeter of the wheel (using the polar coordinate arc length formula) leads to $\int_0^x \sqrt{1+(\partial_t f(t))^2}\, dt = \int_{-\frac{\pi}{2}}^{\theta(x)} \sqrt{r(\alpha)^2 + (\partial_\alpha r(\alpha))^2}\, d\alpha$. We can use the radius condition and differentiation of this equality to get the *arc length condition*: $1 + [r'(\theta(x))\,\theta'(x)]^2 = \left(\frac{d\theta}{dx}\right)^2 \left(r(\theta)^2 + \left(\frac{dr}{d\theta}\right)^2\right)$. And that leads to the simple and elegant differential equation $\frac{d\theta}{dx} = \frac{1}{r(\theta(x))}$, with initial value $\theta(0) = -\frac{\pi}{2}$. This can all be automated using differentiation and Solve as follows.

```
In[∘]:= Solve[∂x (∫_0^x Sqrt[1 + (∂t (-r[θ[t]]))^2] dt == ∫_{-π/2}^{θ[x]} Sqrt[r[α]^2 + (∂α r[α])^2] dα),
  ∂x θ[x], Reals, Assumptions → r[θ[x]] > 0]

Out[∘]= {{θ'[x] → 1/r[θ[x]]}}
```

The road is then $y = -r(\theta(x))$ where $\theta(x)$ is the solution to the differential equation. Because θ is an increasing function of x, it has an inverse $x(\theta)$ which satisfies $\frac{dx}{d\theta} = r(\theta)$; this can be solved by the definite integral $x = \int_{-\pi/2}^{\theta} r(t)\, dt$. The result must then be inverted to get $\theta(x)$. For more on these formulas, and variants in the case that the wheel is given in a different form, see [2, 6, 7, 8, 9, 10].

Here we always take the polar center of the wheel to be $(0, 0)$. It might happen that this point is not inside the wheel. Later we will see the importance of a wheel that is a circle some distance from the origin. Or the wheel might be simply a straight line.

The Catenary Road for a Straight Line

If $r(\theta) = 1$, the wheel is a circle, $\theta(x) = x - \frac{\pi}{2}$, and the road is the straight line $y = -1$. But if $r(\theta) = -\csc\theta$, then the wheel is a straight line (here $-\pi < \theta < 0$) and the road turns out to be the catenary $y = -\cosh x = \frac{1}{2}(e^x + e^{-x})$. The intermediate equations relating θ and x are $\theta = -2\cot^{-1}(e^x)$ and $x = -\ln\left(-\tan\frac{\theta}{2}\right)$. Here is how to solve the differential equation and obtain the road in one line.

```
In[∘]:= ExpToTrig[
  TrigExpand[Csc[θ[x]] /. DSolve[θ'[x] Csc[θ[x]] == -1 && θ[0] == -π/2, θ[x], x][[1]]]]

Out[∘]= -Cosh[x]
```

Here is how to get all the relationships from a solution to the differential equation.

In[∘]:=
```
r[θ_] := -Csc[θ];
θFunc = θ[x] /. DSolve[θ'[x] r[θ[x]] == 1 && θ[0] == -π/2, θ[x], x][[1]]
Simplify[x /. Solve[θ == θFunc, x, ℝ][[1]], -π < θ < 0] /. Log[z_] :> Log[-1/z]
ExpToTrig[TrigExpand[-r[θFunc]]]
```

Out[∘]= $-2\,\mathrm{ArcCot}\left[e^{x}\right]$

Out[∘]= $\mathrm{Log}\left[\mathrm{Tan}\left[\frac{\theta}{2}\right]\right]$

Out[∘]= $-\mathrm{Cosh}[x]$

Here are the same results, using the definite integration formulation (the assumption on ρ is needed for the integration).

In[∘]:=
```
xLine = Assuming[-π/2 ≤ ρ < 0, ∫_{-π/2}^{ρ} r[θ] dθ /. ρ → θ]
θFunc = FullSimplify[θ /. Solve[x == xLine, θ][[1]], x ∈ Reals] /. c_1 → 0
ExpToTrig[TrigExpand[-r[θFunc]]]
```

Out[∘]= $-\mathrm{Log}\left[-\mathrm{Tan}\left[\frac{\theta}{2}\right]\right]$

Out[∘]= $-2\,\mathrm{ArcCot}\left[e^{x}\right]$

Out[∘]= $-\mathrm{Cosh}[x]$

Figure 2 shows an infinite line rolling along a catenary. The "center" of the line — the point $(0, 0)$ — moves along the x-axis. Remarkably, this relationship was discovered by James Clerk Maxwell: in February, 1849 (he was 18 at the time) he published a paper [3] with many examples of rolling wheels. Maxwell’s example 12 [3, p. 537] says

"The straight line whose equation is $r = a \sec \theta$, rolled on a catenary whose parameter is a, traces a line whose distance from the vertex is a."

Out[ ]=



Figure 2. A straight line (polar form $r = -\csc\theta$) rolling along the catenary $y = -\cosh x$: the polar center of the line starts at the origin and follows the x-axis.

Here the line mentioned at the end of the quote is the path of the polar center of the straight line and Maxwell is observing that this center will take a purely horizontal path as the line rolls along a catenary. It took over a century for this idea to be rediscovered by someone (G. B. Robison [6]) who had the idea of truncating the line to a segment, truncating the catenary so that the slope has 45° at the ends, and linking the catenaries so that the rolling object becomes a perfect square.

A Square Wheel

To go from the line-and-catenary to a square wheel one simply (this observation took 111 years) truncates the catenary where its slope is ± 1. At those two points the catenary will make a 45° angle with vertical, and so placing another copy of the truncated catenary beside it gives a 90° receptacle for the corner of a square (Fig. 3). The derivative of $\cosh x$ is $\sinh x$, so this happens at $x = \pm \sinh^{-1} 1$, about ± 0.88.

Out[ ]=



Figure 3. A square rolls on inverted catenaries.

With this template in hand, one can construct the road and a square-wheel device that rides smoothly along the road. Figure 4 shows S.W. on the tricycle at Macalester College; the road is 25 feet long. Figure 5 shows that Ripley's Believe It or Not found this whole idea hard to believe. And Figure 6 shows the round variant built by the National Museum of Mathematics in New York City.

Figure 4. Stan Wagon on his square-wheel bike (it is really a tricycle).

Out[ ]=



Figure 5. The square-wheel tricycle was featured in Ripley's *Believe It or Not*.

Figure 6. The circular version of the square-wheel tricycle at the National Museum of Mathematics in New York City. Photo by National Museum of Mathematics ⟨http://momath.org⟩.

The demonstration below allows one to select a variety of wheel–road combinations. The triangle case is interesting because an equilateral triangle will crash into the road before it falls into the cusp. A workaround for this problem is discussed later in this post.

Out[∘]=



Figure 7. A demonstration that illustrates several road–wheel relationships.

A curiosity in this area, due to G. B. Robison (see [2]) is that there is only one wheel shape for which the road has the exact same shape as the wheel: the road is the parabola defined by $y = -x^2 - \frac{1}{4}$ and the wheel is given by $y = x^2 - \frac{1}{4}$ (see Figure 4). To verify this, one uses the parabola's polar form $r = \frac{1}{2-2\sin\theta}$ and definite integration as discussed earlier. One can see the rolling parabola in the demonstration above.

The Cody Dock Bridge

The preceding discussion is somewhat old news, but something remarkable happened recently. Thomas Randall-Page and Alfred Jacquemot, along with a large team of designers, engineers, and fabricators (design and engineering: Price & Myers; fabrication: Cake Industries) in London, England, completed the installation of a rolling bridge based on the rolling square idea (see Bridge Design). The Cody Dock bridge is anchored on two steel squares and, almost all the time, allows walkers and cyclists to cross the River Lea (Figure 7). But when a boat wishes to pass, a hand crank rolls the bridge along two modified catenaries so that it turns upside-down and the boat can pass below. It can then be returned, again by a hand crank, to its default position. See [1, 4, 5]. The striking features of the bridge are these:

• The bridge weighs 13.2 tons, over 26000 pounds.

• The bridge can be rolled to its upside-down position by a hand crank; there are two cranks, at opposite ends.

• The outside side-length of the two squares is 5.44 meters, or just under 18 feet. The length of the bridge deck is 7.56 meters, or just under 25 feet.

• For this to work, the center of mass at the bridge has to be close to the geometric center. But the missing two edges at the top and the weight of the bridge deck imply that the center of mass is well below the center of the squares. To counterbalance this, 5500 pounds of concrete and scrap steel are hidden in the upper edges of the squares.

• The aforementioned counterweight is chosen so that the center of mass is two inches above the geometric center. This was done to make it clear which winch is doing the work in each direction; if the true center was used, this point might be confused by a strong wind. So when the bridge is being moved into the upside-down position, it is being moved downhill (and one winch acts as a brake), and in the reverse direction it is moving slightly uphill (4 inches in 36 feet, about a 1% grade).

• The two rolling squares are wrapped in oak. This avoids steel-on-steel contact, and the oak can be easily replaced as it wears out. This is also more ecologically sound than plastic or rubber, because this surface will erode and small bits will fall into the water. Currently, one boat passes under the bridge every week, approximately.

• In order to be sure that gravity cannot cause the bridge to slide down the steel road, each square has a series of teeth that bite onto indentations embedded in the road. If the square were true squares, there would not be enough room for these teeth near the cusp. Also, as we learned from our tricycle and its rubber rims, there is additional wear on the corners when they are true right angles. To get around these concerns, the corners of the squares are rounded, using quarter-circles of radius about an eighth of the square’s side length. This adds some interesting complexity to the shape of the road.

Figure 7. The Cody Dock bridge rolling along catenaries toward its upside-down position along. Photo by G. G. Archard.

Out[∘]=

Figure 8. The entire rolling sequence through 180°.

The rounding of the corners leads to some interesting new mathematics. Assume the square has side-length 2 and is centered at the origin. So the polar form of a side is $r = -\csc\theta$. Let b be the radius of the rounded corner (about

0.25 in the actual bridge; see Figure 8). Let $\bar{b} = 1 - b$. Then some elementary geometry shows that the corner has the polar form $r = \sqrt{2}\,\bar{b}\cos\left(\frac{\pi}{4}+\theta\right) + \sqrt{b^2 - 2\,\bar{b}^2\sin\left(\frac{\pi}{4}+\theta\right)^2}$.

Out[∘]=



Figure 8. Rounding the corners using the polar form $r = \sqrt{2}\,\bar{b}\cos\left(\frac{\pi}{4}+\theta\right) + \sqrt{b^2 - 2\,\bar{b}^2\sin\left(\frac{\pi}{4}+\theta\right)^2}$ (with θ translated by multiples of $\frac{\pi}{2}$ for the four corners). Here $b = 0.25$. The wheel at right comes from single piecewise polar plot.

In general, one might use a numerical approach to the differential equation and `NDSolve` makes that fairly simple (see [9], where the user can change the shape of the road using locators). But it is interesting to see if a symbolic solution exists, and in this case it does, and we can use the definite integration approach. The road starts with the familiar catenary, $y = -\cosh x$, to handle the straight part of the wheel. But when we come to the rounded corner, some new work is needed. The approach via a definite integral works and yields a symbolic result using the complete elliptic integral of the second kind, $E(\phi\,|\,m)$, which is $\int_0^\phi \sqrt{1 - m\sin^2 t}\; dt$; one can use this definition directly to construct the road (that was also done by Alfred Jacquemot of the design team). We use $B(b)$ for $\cot^{-1}\bar{b}$. We have $x = \int_{-B(b)}^{\theta} r(\rho)\, d\rho$. The lower limit is because $x(-B(b)) = 0$, a consequence of where the corner starts. This definite integral evaluates more smoothly when the lower limit of integration and also some coefficients, are left undefined. In this call to `Integrate`, BB represents $B(b)$.

In[∘]:=
```
SetOptions[Integrate, GenerateConditions → False];
```

$$\int_{-\mathrm{BB}}^{\theta}\left(\alpha\,\mathrm{Cos}\left[\frac{\pi}{4}+\rho\right] + \mathrm{b}\,\sqrt{1-\beta^2\,\mathrm{Sin}\left[\rho+\frac{\pi}{4}\right]^2}\right) d\rho \;/.\; \left\{\alpha \to \sqrt{2}\,(1-\mathrm{b}),\ \beta \to \sqrt{2}\,\frac{1-\mathrm{b}}{\mathrm{b}}\right\}$$

Out[∘]=

$$\mathrm{b}\left(\mathrm{EllipticE}\left[\mathrm{BB}-\frac{\pi}{4},\ \frac{2\,(1-\mathrm{b})^2}{\mathrm{b}^2}\right] + \mathrm{EllipticE}\left[\frac{\pi}{4}+\theta,\ \frac{2\,(1-\mathrm{b})^2}{\mathrm{b}^2}\right]\right) + (1-\mathrm{b})\,(-\mathrm{Cos}[\mathrm{BB}] + \mathrm{Cos}[\theta] + \mathrm{Sin}[\mathrm{BB}] + \mathrm{Sin}[\theta])$$

Replacing BB by $B(b)$, which is $\cot^{-1}\bar{b}$ and simplifying gives

$$x = b\left(E\left(\theta+\frac{\pi}{4}\,\middle|\,\frac{2\,\bar{b}^2}{b^2}\right) + E\left(B(b)-\frac{\pi}{4}\,\middle|\,\frac{2\,\bar{b}^2}{b^2}\right)\right) + \bar{b}\,(\cos\theta+\sin\theta) + \frac{b\,\bar{b}}{\sqrt{1+\bar{b}^2}}.$$

Now, the road defined by this equation has to be translated to the right by $\sinh^{-1}(\bar{b})$ because it starts at that value of x. And the pieces can be extended using periodicity to give a single road function $(x, -r(\theta(x)))$ that can be plotted in parametric form. Note that we cannot solve $x = x(\theta)$ for θ to get the road in the form $y = f(x)$.

Out[ ]=



Figure 9. The almost-square wheel and the two-part road (catenary in blue, and an elliptic function graph in orange) needed to accommodate it.

The following demonstration allows one to change the corner radius. A radius of 0 gives a rolling square while a radius of 1 shows the traditional rolling circle.



Figure 10. This demonstration allows one to change the amount of rounding at the corners of the squae.

One can use the various formulas to get the physical form of the bridge, using solid shapes for the wheels. Figure 11 shows the 3-dimensional bridge, at rest and nearing the upside-down position.

In[∘]:=
```
Show[CodyDockBridgeRealistic3D[0.25, 0], wheelImageRealisticRotated3D[0.25, 2.55],
  ImageSize → 800, ViewPoint → {0.5, -4.5, 0.8}, Lighting → "Accent"]
```

Out[∘]=



Figure 11. A 3-dimensional model showing how the Cody Dock Bridge moves to an upside-down position so that a boat can pass underneath it.

Triangular Difficulties

A triangle is even farther from a circle than a square, though the square holds a lock as the epitome of a noncircle. But the triangle is extremely interesting from the rolling polygon perspective. As discovered by Hall and Wagon [2], a triangle cannot physically roll on the appropriate catenaries because a corner crashes into the road before it lands in the cusp (Figure 12). But, borrowing an idea from the Cody Dock bridge, we can get a rolling triangle if we round the corners a little bit. If the triangle side measures two units, then using a rounding radius of 0.084 is adequate to eliminate the collision. Note that the triangle faces a steeper slope than the square does, so for a physical model one must choose materials that prevent slippage when the contact point is at a place where the catenary is steep.

The rounded corners of a triangular wheel are not easily expressed in polar form $r = r(\theta)$. This is because for a circle in a general position, the relation between r and θ is a quadratic equation in r. (The case of a square wheel is special, because the centers of the circles lie on the lines $y = \pm x$ and this leads to a considerable simplification.) So it is easier to represent the triangular wheel as a parametric curve $g(t) = (g_1(t), g_2(t))$. As before, the center of the wheel is $(0, 0)$. The parametrization of the road on which it will roll is then (see [2])

$$x(t) = \int_0^t \frac{g_1(s)\, g_2'(s) - g_1'(s)\, g_2(s)}{\sqrt{g_1(s)^2 + g_2(s)^2}}\, ds \qquad y(t) = -\sqrt{g_1(t)^2 + g_2(t)^2}\,.$$

We use `NDSolve` to calculate the first integral. Finally, the rotation of the wheel when it touches the point $(x(t), y(t))$ is obtained by measuring the angle between the line segment from $(0, 0)$ to $g(t)$ and the vertical direction.

In[1169]:=
```
Row[Show[singleImage[3, #1, 0.4], ImageSize → 300,
   PlotRange → {{-0.8, 1.45}, {-1.2, 0.5}}] & /@ {0, 0.09}]
```



Figure 12. Left: A triangle cannot roll because there is a serious collision between a vertex and the catenary road. Right: Rounding the corners by 5% (the rounding radius as a percentage of the side length) eliminates the collision.

The demonstration that follows handles rolling n-gons for $n = 3, 4, 5$, or 6.

In[35]:=
```
demoTriangle
```

Out[35]=



Figure 13. A demonstration for rolling triangles, squares, pentagons, and hexagons.

Conclusion: The Perfect Catenary

Catenaries have been used in bridge construction for a long time. The shape forms the strongest arch, and so one often sees such curves in the arches supporting a bridge from below. And a catenary is the natural shape of a cable, so the curve occurs very often in the cables from which a bridge hangs. It is wonderful to see a completely new use of catenaries in bridge construction: lying below and to the side of the bridge and allowing the bridge to be turned upside-down. Further, the traditional appearance of catenaries is imperfect: the weight of the bridge deck means that the shapes that arise are not exact catenaries. But the Cody Dock bridge uses a perfect catenary for the surface on which the straight part of the bridge rolls. What a remarkable design!

References

1. K. Bokhammas, Ready to roll, *Bridge Design and Engineering*, No. 109, 2022, 48–49, and cover. Also https://www.bridgesawards.co.uk/winners/winners-2023 .

2. L. Hall and S. Wagon, *Roads and wheels*," *Mathematics Magazine*, 65(5), 1992 pp. 283–301.

3. J. C. Maxwell, On the theory of rolling curves, *Transactions of the Royal Society of Edinburgh*, 16(5), 1849 pp. 519–540 (p. 537, Ex. 12). archive.org/details/transactionsofro16roy/page/518/mode/2up .

4. Thomas Randall-Page, Cody Dock Rolling Bridge https://thomasrandallpage.com/Cody-Dock-Rolling-Bridge .

5. Thomas Randall-Page, Cake Industries, Price & Myers, Cody Dock Rolling Bridge: Fabrication and installation, https://www.youtube.com/watch?v=M_HTNX2M3X8 .

6. G. B. Robison, Rockers and rollers, *Mathematics Magazine*, 33(3), 1960 pp. 139–144. (Result first in E1033 1952.) www.jstor.org/stable/3029034?origin=crossref .

7. A. Slavik, S. Wagon, and D. Schwalbe, *VisualDSolve: Visualizing Differential Equations with Mathematica*, 2nd ed., Wolfram Research, Inc., 2009, Chapter 12, http://www.wolfram.com/books/profile.cgi?id=9553 .

8. S. Wagon, The ultimate flat tire, *Math Horizons*, (Feb. 1999) 14–17.

9. S. Wagon, Shaping a road and finding the corresponding wheel, Wolfram Demonstrations Project, 2011, http://demonstrations.wolfram.com/ShapingARoadAndFindingTheCorrespondingWheel

10. D. G. Wilson, Problem E1668, *American Mathematical Monthly*, 71(2), 1964 p. 205. Solution, 72(1), 1965 pp. 82–83.

11. The Bridges Design Award, https://www.bridgesawards.co.uk/winners/winners-2023 .